
Bedienung 

Programm starten (Sprung nach 0600, ROM C) 
0420 C4 05  LDI #05 
0422 35  XPAH1 
0423 C4 FF  LDI #FF 
0425 31  XPAL1 

P1=05FF 

0426 3D  XPPC1 Sprung nach 0600 
[GO] (Anzeige bleibt unverändert) 

Alternativ Sprung nach 0600 nur bei aktiver SB Taste, als Anfang des Anwenderprogramms 
0420 06  CSA 
0421 D4 20  ANI #20 
0423 98 07  JZ ANWPRG 

Sprung zum Anwenderprogramm bei 
042C falls SENSE B bei GO nicht 
gedrückt 

0425 C4 05  LDI #05 
0427 35  XPAH1 
0428 C4 FF  LDI #FF 
042A 31  XPAL1 
042B 3D  XPPC1 

Sprung ins Kassetten-Programm bei 
aktiver SENSE B Taste 

[GO] mit oder ohne SENSE B (Anzeige bleibt unverändert) 
042C xx xx ANWPRG: xxxx Start Anwenderprogramm 

 
Eingabe der Start- und Endadresse 

041C xx   Startadresse HOB 
041D xx   Startadresse LOB 
041E xx   Endadresse HOB 
041F xx   Endadresse LOB 

 
Tastenfunktionen im Kassettenprogramm 
0..F, AD, ME, IN Für Eingabe von Start- und Endadressen, solange Laden oder Speichern noch 

nicht gestartet ist. Funktion analog Betriebsprogramm. 
AC Schiebt Speicherblock von (StartAdresse) bis (Endadresse) um eine Adresse 

nach oben. Start und Endadresse werden inkrementiert 
GO Auf Kassette speichern, Synchronisations-Signal 2.4kHz wird ausgegeben. 

Etwa 5 Sekunden aufnehmen. SENSE A startet Ausgabe der Daten, F0 zeigt 
diese an. Nach Beenden der Aufnahme (F0 bleibt statisch), RESET drücken. 
Vor dem Speichern ev. die ersten 7 Bytes (benutzt durch ROM-Einsprung) 
wieder herstellen oder Alternativ-Methode (siehe oben) benutzen.  

SI Von Kassette laden. SI erst drücken, wenn Synchronisationssignal  von 
Kassette anliegt. SOUT zeigt Daten. Nach erfolgreichem Laden Anzeige 0420 
xx, bei Checksummen-Fehler EEEE EE. Der Fehler tritt auch auf, wenn Start- 
und Endadressen nicht mit der geladenen Länge übereinstimmen. 

Alle anderen Sprung nach 0000, Reset, Programm wird beendet 

Scratchpad 

0410 10 ISEDH  Eingabe SED HOB 
0411 11 ISEDL  Eingabe SED LOB 
0412 12 DSADH  Anzeige Adresse HOB 
0413 13 DSADL  Anzeige Adresse LOB 
0414 14 DSCOD  Anzeige Code 
0415 15 DEBO1  Entprellen1 
0416 16 SPKEY  Funktionstaste 
0417 17 DEBO2  Entprellen2 
041B 1B KYLI4  Tasten-Zeilen4 
041C 1C FRADH  Start Adresse HOB 
041D 1D FRADL  Start Adresse LOB 
041E 1E TOADH  End Adresse HOB 
041F 1F TOADL  End Adresse LOB 

Einsprung Kassetteninterface 

0600 C4 01 START: LDI #01 P3=010F 
0602 37  XPAH3  
0603 C4 0F  LDI #0F  
0605 33  XPAL3  
0606 3F  XPPC3 Sprung nach 0110 (ROM A, EINAU) 

 
Auf Funktionstaste warten 

0607 C4 04  LDI #04 P1=0400 (Pointer Scratchpad) 
0609 35  XPAH1  
060A C4 00  LDI #00  
060C 31  XPAL1  
060D C1 16  LD SPKEY(1) A=Code Funktionstaste 
060F 98 EF  JZ START Keine, erneut lesen 

 
Funktionstaste auswerten 

0611 01   XAE EX=Tastencode, 01-0E 
0612 C4 06  LDI #06 P2=0622, JMPTAB-1 
0614 36  XPAH2  
0615 C4 22  LDI #22  
0617 32  XPAL2  
0618 C2 80  LD 80(2) A=P2(EX), HOB Sprungziel 
061A 37  XPAH3  
061B C6 0F  LD @0F(2) P2 + 0F (zeigt auf LOB des Sprungziels) 
061D C2 80  LD 80(2) A=P2(EX), LOB Sprungziel 
061F 33  XPAL3  
0620 3F  XPPC3 Sprung je nach Taste 
     

0621 90 DD  JMP START Endlosschleife 
 
Tasten-Sprungtabelle 

0623 00 JMPTAB:  HOB Taste 01: Adresse = 0000 (PC) 
0624 00   HOB Taste 02: Adresse = 0000 (P1) 
0625 00   HOB Taste 03: Adresse = 0000 (P2) 
0626 00   HOB Taste 04: Adresse = 0000 (P3) 
0627 07   HOB Taste 05: Adresse = 07BF (AC) 
0628 00   HOB Taste 06: Adresse = 0000 (EX) 
0629 00   HOB Taste 07: Adresse = 0000 (BP) 
062A 00   HOB Taste 08: Adresse = 0000 (0801) 
062B 00   HOB Taste 09: Adresse = 0000 (PT) 
062C 06   HOB Taste 0A: Adresse = 068F (SI) 
062D 07   HOB Taste 0B: Adresse = 0707 (GO) 
062E 06   HOB Taste 0C: Adresse = 0656 (AD) 
062F 06   HOB Taste 0D: Adresse = 063E (ME) 
0630 06   HOB Taste 0E: Adresse = 067D (IN) 
0631 00   LOB Taste 01: Adresse = 0000 (PC) 
0632 00   LOB Taste 02: Adresse = 0000 (P1) 
0633 00   LOB Taste 03: Adresse = 0000 (P2) 
0634 00   LOB Taste 04: Adresse = 0000 (P3) 
0635 BF   LOB Taste 05: Adresse = 07BF (AC) 
0636 00   LOB Taste 06: Adresse = 0000 (EX) 
0637 00   LOB Taste 07: Adresse = 0000 (BP) 
0638 00   LOB Taste 08: Adresse = 0000 (0801) 
0639 00   LOB Taste 09: Adresse = 0000 (PT) 
063A 8F   LOB Taste 0A: Adresse = 068F (SI) 
063B 07   LOB Taste 0B: Adresse = 0707 (GO) 
063C 56   LOB Taste 0C: Adresse = 0656 (AD) 

063D 3E   LOB Taste 0D: Adresse = 063E (ME) 
063E 7D   LOB Taste 0E: Adresse = 067D (IN) 

Taste ME: Nächste Adresse anzeigen 

063F 02 KEYME: CCL CY=0 
0640 C1 13  LD DSADL(1) A=(Anzeige Adresse LOB) 
0642 F4 01  ADI #01 A=A+1+CY 
0644 C9 13  ST DSADL(1) A->(Anzeige Adresse LOB) 
0646 C1 12  LD DSADH(1) A=(Anzeige Adresse HOB) 
0648 F4 00  ADI #00 A=A+CY 
064A C9 12  ST DSADH(1) A->(Anzeige Adresse HOB) 

 
064C C1 13 L064C: LD DSADL(1) A=(Anzeige Adresse LOB) 
044E 32  XPAL2  
044F C1 12  LD DSADH(1) A=(Anzeige Adresse HOB) 
0651 36  XPAH2  
0652 C2 00  LD 00(2) A=(Anzeige Adresse) 
0654 C9 14  ST DSCOD(1) A->(Anzeige Code) 
0656 3F  XPPC3 Return 

Taste AD: Vorherige oder eingegebene Adresse anzeigen 

0657 C1 10 KEYAD: LD ISEDH(1) A=(Eingabe SED HOB) 
0659 9C 04  JZ L065F  
065B C1 11  LD ISEDL(1) A=(Eingabe SED LOB) 
065D 9A 10  JZ 10(2)  

 
065F C1 11 L065F: LD ISEDL(1) A=(Eingabe SED LOB) 
0661 C9 13  ST DSADL(1) A->(Anzeige Adresse LOB) 
0663 C1 10  LD ISEDH(1) A=(Eingabe SED HOB) 
0665 C9 12  ST DSADH(1) A->(Anzeige Adresse HOB) 

 
0667 C4 00 L0667: LDI #00  
0669 C9 11  ST ISEDL(1) 0->(Eingabe SED LOB) 
066B C9 10  ST ISEDH(1) 0->(Eingabe SED HOB) 
066D 90 DD  JMP L064C P2=(Anzeige Adresse), Code anzeigen 

 
066F 03  SCL CY=1 
0670 C1 13  LD DSADL(1) A=(Anzeige Adresse LOB) 
0672 FC 01  CAI 01 A+= /01 + CY 
0674 C9 13  ST DSADL(1) A->(Anzeige Adresse LOB) 
0676 C1 12  LD DSADH(1) A=(Anzeige Adresse HOB)  
0678 FC 00  CAI 00 A+= /00 +CY 
067A C9 12  ST DSADH(1) A->(Anzeige Adresse HOB) 
067C 90 CE  JMP L064C P2=(Anzeige Adresse), Code anzeigen 

Taste IN: Eingegebenen Code speichern 

067E C1 12 KEYIN: LD DSADH(1) P2=(Anzeige Adresse) 
0680 36  XPAH2  
0681 C1 13  LD DSADL(1)  
0683 32  XPAL2  
0684 C1 11  LD ISEDL(1) A=(Eingabe SED LOB) 
0686 CE 01  ST @01(2) A->(P2), P2++ 
0688 36  XPAH2  
0689 C9 12  ST DSADH(1) A->(Anzeige Adresse HOB) 
068B 32  XPAL2  

068C C9 13  ST DSADL(1) A->(Anzeige Adresse LOB) 
068E 90 D7  JMP L0667  

Taste SI: Von Kassette laden 

0690 02 KEYSI: CCL CY=0 
0691 C1 1C  LD FRADH(1) P2=Startadresse 
0693 36  XPAH2  
0694 C1 1D  LD FRADL(1)  
0696 32  XPAL2  

 
0697 C4 07 RDBYTE: LDI #07 A=7, Bit-Zähler 
0699 C9 1B  ST KYLI4(1) A->Tasten-Zeilen4 (BitZähler) 

 
Auf Startbit warten (von B12 Interface Print) 

069B 06 WSENSB: CSA (ST)->(AC) inkl. SENSE A, B 
069C D4 20  ANI #20 SENSE B (Ausgang Interface) ? 
069E 9C FB  JNZ WSENSB    Ja, warten  

 
06A0 C4 90  LDI #90 A=90 
06A2 8F 01  DLY 01  

 
Ein Byte einlesen (über SIN, von A7 Interface Print) 

06A4 C4 46 BITLOOP: LDI #46 A=46 
06A6 8F 03  DLY 03  
06A8 19  SIO SIN -> EX bit7, EX bit0 -> SOUT 
06A9 B9 1B  DLD KYLI4(1) A=(Tasten-Zeilen4) -1 (BitZähler) 
06AB 94 F7  JP BITLOOP  

 
Gelesenes Byte speichern 

06AD 40  LDE A=EX, an SIN gelesenes Byte 
06AE CA 00  ST 00(2) A->(Startadresse) 

 
Endadresse erreicht? 

06B0 36  XPAH2  
06B1 01  XAE (AC)<->(EX) 
06B2 40  LDE  
06B3 36  XPAH2  
06B4 40  LDE  
06B5 E1 1E  XOR TOADH(1) A^=(Endadresse HOB) 
06B7 9C 09  JNZ RDNEXT HOB ungleich, nächstes Byte lesen  

 
06B9 32  XPAL2  
06BA 01  XAE  
06BB 40  LDE  
06BC 32  XPAL2  
06BD 40  LDE  
06BE E1 1F  XOR TOADL(1) A^=(Endadresse LOB) 
06C0 98 06  JZ RDDONE HOB und LOB gleich, fertig 

 
Noch nicht fertig, Startadresse inkrementieren und nächstes Byte lesen 

06C2 8F 04 RDNEXT: DLY 04  
06C4 C6 01  LD @01(2) (Startadresse++) A=irrelevant 
06C6 90 CF  JMP RDBYTE  
     

Endadresse erreicht 
06C8 C4 00 RDDONE: LDI #00 A=0 
06CA C9 17  ST DEBO2(1) 0->(Entprellen2, Checksummenzähler) 

06CC C1 1C  LD FRADH(1) P2=Startadresse 
06CE 36  XPAH2  
06CF C1 1D  LD FRADL(1)  
06D1 32  XPAL2  
     

Checksumme berechnen, 8-Bit, alle Bytes addieren, ohne Uebertrag 
06D2 C6 01 CSLOOP: LD @01(2) A=(P2++) 
06D4 F1 17  ADD DEBO2(1) Checksumme += nächstes Byte 
06D6 C9 17  ST DEBO2(1)  
     

06D8 36  XPAH2 HOB Start- und Endadresse vergleichen 
06D9 01  XAE  
06DA 40  LDE  
06DB 36  XPAH2  
06DC 40  LDE  
06DD E1 1E  XOR TOADH(1) A^=(Endadresse HOB) 
06DF 9C F1  JNZ CSLOOP  

 
06E1 32  XPAL2 LOB Start- und Endadresse vergleichen 
06E2 01  XAE  
06E3 40  LDE  
06E4 32  XPAL2  
06E5 40  LDE  
06E6 E1 1F  XOR TOADL(1) A^=(Endadresse LOB) 
06E8 9C E8  JNZ CSLOOP  

 
Checksumme prüfen 

06EA C2 00  LD 00(2) A= letztes gelesenes Byte 
06EC E1 17  XOR DEBO2(1) A^=(Entprellen2) 
06EE 9C 07  JNZ CSERR  
     

Checksumme ok, Sprung nach 0000, Programm beenden 
06F0 C4 00 Reset: LDI #00 P3=0000 
06F2 33  XPAL3  
06F3 C4 00  LDI #00  
06F5 37  XPAH3  
06F6 3F  XPPC3 Soft-Reset 
     

Checksummenfehler, Anzeige 'EEEE EE' 
06F7 C4 EE CSERR: LDI #EE A=EE 
06F9 C9 12  ST DSADH(1) A->(Anzeige Addresse HOB) 
06FB C9 13  ST DSADL(1) A->(Anzeige Addresse LOB) 
06FD C9 14  ST DSCOD(1) A->(Anzeige Code) 
     

06FF C4 01 ERRLOOP: LDI #01 P3=010F 
0701 37  XPAH3  
0702 C4 0F  LDI #0F  
0704 33  XPAL3  
0705 3F  XPPC3 Weiter bei 0110 (ROM A) 
     

0706 90 F7  JMP ERRLOOP Endlos-Schleife, Reset Taste  

Taste GO: Auf Kassette speichern 

0708 02 KEYGO: CCL CY=0 
0709 C1 1C  LD FRADH(1) P2=Startadresse 
070B 36  XPAH2  
070C C1 1D  LD FRADL(1)  

070E 32  XPAL2  
 
Checksumme über Addressbereich berechnen 

070F C4 00  LDI #00 A=0 
0711 C9 17  ST DEBO2(1) A->(Entprellen2) 

 
0713 C6 01 WRCS: LD @01(2) A=(P2++) 
0715 F1 17  ADD DEBO2(1) A+=(Entprellen2) + CY 
0717 C9 17  ST DEBO2(1) A->(Entprellen2) 
     

0719 36  XPAH2  
071A 01  XAE  
071B 40  LDE  
071C 36  XPAH2  
071D 40  LDE  
071E E1 1E  XOR TOADH(1) A^=(Endadresse HOB) 
0720 9C F1  JNZ WRCS  

 
0722 32  XPAL2  
0723 01  XAE  
0724 40  LDE  
0725 32  XPAL2  
0726 40  LDE  
0727 E1 1F  XOR TOADL(1) A^=(Endadresse LOB) 
0729 9C E8  JNZ WRCS  

 
Checksumme an Endadresse+1 speichern 

072B C1 17  LD DEBO2(1) A=(Entprellen2), Checksumme 
072D CE 01  ST @01(2) A->(Endadresse ++) 
     

Endadresse += 1 
072F 36  XPAH2  
0730 C9 1E  ST TOADH(1) A->(Endadresse HOB) 
0732 32  XPAL2  
0733 C9 1F  ST TOADL(1) A->(Endadresse LOB) 

 
0735 C1 1C  LD FRADH(1) P2=Startadresse 
0737 36  XPAH2  
0738 C1 1D  LD FRADL(1)  
073A 32  XPAL2  

 
2.4 kHz Signal an F1 ausgeben, auf SA Tastendruck warten 

073B 06 WRSYNC: CSA (ST)->(AC), inkl SENSE A, B 
073C F4 02  ADI #02 A+=2 + CY 
073E D4 03  ANI #03  
0740 07  CAS (AC)->(ST), ohne SENSE A, B 
0741 C4 0E  LDI #0E  
0743 8F 00  DLY 00 180uS Verzögerung 

 
0745 06 WRWSA: CSA 
0746 D4 10  ANI #10 
0748 98 F1  JZ WRSYNC 

Auf Taste SA warten. Wenn gedrückt, 
weiter zu Ausgabe Startbit 

 
Startbit und nächstes Byte ausgeben   

074A C4 0F WRBYTE: LDI #0F Bit-Flankenzähler setzen (16) 
074C C9 17  ST DEBO2(1) Entprellen2=Flankenzähler=0F 
074E C4 01  LDI #01  
0750 01  XAE EX=1 (für Startbit) 

Startbit (H) ausgeben   

0751 06 WRSBIT: CSA A=ST inkl SENSEs 
0752 70  ADE A+=EX + CY 
0753 D4 03  ANI #03 F0 & F1 maskieren 
0755 07  CAS Ausgabe an F1, B28 Interface Print 
0756 C4 0D  LDI #0D  
0758 8F 00  DLY 00 175uS Verzögerung 
075A B9 17  DLD DEBO2(1) A=(Entprellen2 --) 
075C 94 F3  JP WRSBIT  

 
Nächstes Byte ausgeben  

075E C4 0B  LDI #0B A=0B 
0760 C9 15  ST DEBO1(1) A->(Entprellen1) 
0762 C6 01  LD @01(2) A=(Startadresse++), nächstes Byte 
0764 37  XPAH3 P3H=nächstes Byte 

 
Nächstes Bit ausgeben (Detailfunktion z.T. unklar)  

0765 37 WRBITS: XPAH3 A=aktuelles Byte, rotiert wenn nicht 1. 
0766 03  SCL CY=1 
0767 1F  RRL A(7)=CY, A>>1, CY=A(0) 
0768 37  XPAH3 P3H=rotiertes Byte, A=irrelevant 
0769 06  CSA A=ST, inkl SENSEs 
076A 33  XPAL3 P3L=ST 
076B 06  CSA A=ST inkl SENSEs 
076C 02  CCL CY=0 
076D 70  ADE A=ST+EX (CY=0) 
076E D4 03  ANI #03 F0 & F1 maskieren 
0770 08  NOP  
0771 01  XAE EX=F0 & F1 
0772 33  XPAL3 A=ST @ 076A 
0773 D4 80  ANI #80 CY/L maskieren 
0775 01  XAE EX=CY/L, A=irrelevant 
0776 06  CSA A=ST, inkl SENSEs 
0777 58  ORE A|=CY/L 
0778 07  CAS ST=A, ohne SENSEs 
0779 07  CAS ST=A, ohne SENSEs  (2x ??) 
077A C4 00  LDI #00 A=0 
077C F4 01  ADI 01 A=1 + CY @ 0777 
077E 01  XAE EX=1 + CY @ 0777, A=irrelevant 
077F 06  CSA A=ST, inkl SENSEs 
0780 70  ADE A+=EX + CY 
0781 D4 03  ANI #03 F0 & F1 maskieren 
0783 08  NOP Timing?? 
0784 07  CAS F0 & F1 ausgeben 

 
0785 C4 0E  LDI #0E A=0E 
0787 C9 17  ST DEBO2(1) A->(Entprellen2) 
0789 B9 15  DLD DEBO1(1) A=(Entprellen1--) 
078B 98 11  JZ WRBDONE Byte fertig, Test ob noch mehr 
078D 90 04  JMP WRBIT Byte noch nicht fertig, nächstes Bit 

 
Bit ausgeben  

078F C4 07 WRBITDLY: LDI #07  
0791 8F 00  DLY 00 Verzögerung 
     

0793 06 WRBIT: CSA AC=ST, inkl SENSEs 
0794 70  ADE EX=0: F1 immer low 

0795 D4 03  ANI #03 F0 & F1 maskieren 
0797 07  CAS F0 & F1 ausgeben 
0798 B9 17  DLD DEBO2(1) A=(Entprellen2--) 
079A 98 C9  JZ WRBITS Bit fertig, nächstes Bit 
079C 90 F1  JMP WRBITDLY  

 
Test ob alles gespeichert 

079E 36 WRBDONE: XPAH2  
079F 01  XAE  
07A0 40  LDE  
07A1 36  XPAH2  
07A2 40  LDE  
07A3 E1 1E  XOR TOADH(1) A^=(Endadresse HOB) 
07A5 9C A3  JNZ WRBYTE  
     

07A7 32  XPAL2  
07A8 01  XAE  
07A9 40  LDE  
07AA 32  XPAL2  
07AB 40  LDE  
07AC E1 1F  XOR TOADL(1) A^=(Endadresse LOB) 
07AE 9C 9A  JNZ WRBYTE  

 
07B0 C1 1F  LD TOADL(1) A=(Endadresse LOB) 
07B2 02  CCL CY=0 
07B3 F4 FF  ADI #FF  
07B5 C9 1F  ST TOADL(1) A->Endadresse LOB 
07B7 C1 1E  LD TOADH(1) A=Endadresse HOB 
07B9 F4 FF  ADI #FF  
07BB C9 1E  ST TOADH(1) A->Endadresse HOB 
07BD 90 86  JMP WRWSA Wieder Sync augeben und auf SA warten 
07BF 00  HALT  

Taste AC: Speicherblock um 1 Adresse nach oben schieben  

07C0 C4 04 KEYAC: LDI #04 P1=0420 
07C2 35  XPAH1  
07C3 C4 20  LDI #20  
07C5 31  XPAL1  
07C6 C1 FE  LD FE(1) A=(041E), (Endadresse HOB) 
07C8 36  XPAH2  
07C9 C1 FF  LD FF(1) A=(041F), (Endadresse LOB) 
07CB 32  XPAL2 P2=(Endadresse) 

 
07CC C6 FF L07CC: LD @FF(2) A=(--Endadresse) 
07CE CA 01  ST 01(2) A->(Endadresse+1) 
07D0 36  XPAH2 A=Endadresse HOB 
07D1 01  XAE A<->EX 
07D2 40  LDE A=EX=Endadresse HOB 
07D3 36  XPAH2 P2H=Endadresse HOB 
07D4 40  LDE A=EX=Endadresse HOB 
07D5 E1 FC  XOR FC(1) A^(Startadresse HOB) 
07D7 9C F3  JNZ L07CC  

 
07D9 32  XPAL2  
07DA 01  XAE  
07DB 40  LDE  
07DC 32  XPAL2  

07DD 40  LDE  
07DE E1 FD  XOR FD(1) A^(Startadresse LOB) 
07E0 9C EA  JNZ L07CC  

 
07E2 CA 00  ST 00(2)  
07E4 C4 06  LDI #06 P2=0600 (Sprungaddresse für L07F6) 
07E6 36  XPAH2  
07E7 C4 00  LDI #00  
07E9 32  XPAL2  
07EA A9 FD  ILD FD(1) (Startadresse LOB ++) 
07EC 9C 02  JNZ L07F0 Wenn<>0, Startadr HOB unverändert 

 
07EE A9 FC  ILD FC(1) (Startadresse HOB ++) 
07F0 A9 FF L07F0: ILD FF(1) (Endadresse LOB ++) 
07F2 9C 02  JNZ L07F6 Wenn <>0, Endaddr HOB unverändert 

 
07F4 A9 FE  ILD FE(1) (Endadresse HOB ++) 
07F6 92 FF L07F6: JMP FF(2) Sprung an Anfang (0600) 

 
07F8 00  HALT  
07F9 00  HALT  
07FA 00  HALT  
07FB 00  HALT  
07FC 00  HALT  
07FD 00  HALT  
07FE 00  HALT  
07FF 00  HALT  

